# Sources of 2016 Premium Increases


## Seth J. Chandler
## Foundation Professor of Law
## University of Houston Law Center


The views expressed herein do not necessarily represent those of the University of Houston.

---

## Introduction

My name is Seth Chandler and I am a law professor at the University of Houston with specialization in insurance law and health care law. I also do a lot of work using mathematics to enhance legal analysis. I am here to testify on some of the sources of anticipated premium increases on the Exchanges for 2016 with an emphasis on two of the "3 Rs."

Here is a summary of what I have to say:

- We need to be careful in looking at premium increases: there are many occasions on which the net premium percent increase seen by an insured, the important thing, will be considerably higher than the gross premium percent increase. This fact is likely to diminish individual choice and induce policyholders to purchase lower cost silver HMO policies. Sometimes the net premium increase will be less than the gross.
- Although they will contribute, the phase out of transitional reinsurance and the Cromnibus alteration of risk corridors are unlikely to be responsible for particularly large premium increases for 2016.
- The major source of increases is likely to be higher-than-expected claims from insureds, particularly in the more generous platinum and gold plans.

Because I have been obliged to write this before King v. Burwell is resolved, I am going to assume that the case is resolved favorably to the Obama administration' s position. If that is not the case, the issues created thereby will not make what I say untrue in concept, but the numbers may well change and there will be an issue of greater magnitude to debate.

---

## Net premiums will often rise more than gross premiums, particularly for low income policyholders

The gross premium increases that may be coming are troubling to the stability of the ACA. What should be yet more worrisome, however, is the increase in *net* premiums chronically ill or lower income purchasers are likely to see. This is because gross premiums will not determine most consumer's behavior: net premiums, the amount paid after lawful subsidies are taken into account are what will matter. And, as I show in a technical appendix to my written testimony, the rate of net premium increases is not the same as the rate of gross premium increases. Rather, the rate of net premium increases are the difference between the gross premium increases divided by the prior years net premium. As the denominator of that fraction decreases -- as the person gets poorer -- the net premium increase grows. I show in the appendix how this fact can easily convert a 10% gross premium increase into a 15% gross premium index. Or how it can convert a 10% gross premium increase into a 12% net premium increase in a way that may impel the purchaser to experience a 50% increase in out of pocket costs.

This is not some bug in the ACA. It is a feature. Baked right into the architecture. What it means, however, if higher metal level policy premiums rise faster than silver policies and or PPOs rise faster than HMOs, is that there will be more pressure than one might expect for purchasers to head for silver HMOs.

---

# The phasing out of transitional reinsurance should cause significant but not enormous increases in premiums for most insurers

Part of the Affordable Care Act was to provide insurers participating in the Exchanges for free with something they otherwise might have purchased: specific stop loss reinsurance. This “R” reduced risk a bit for smaller insurers but, more importantly, permitted insurers to offer lower gross premiums for all purchasers. Unlike the cost sharing subsidies and tax credits, Congress chose by this plan to make the reduction no greater for poorer purchasers than for the wealthier.

The estimates I am to provide are based on the Obama administration' s "Actuarial Value Calculator" for 2014, 2015 and the current draft for 2016. These calculators have behind them estimates for the distribution of claims expenses called continuance tables for purchases of each of the four metal levels of policies. It is the method by which CMS determines whether a policy is really offering benefits equal to some specified percentage of claims expenses, 70% for silver, 80% for gold, etc. Thus, I would certainly hope it is accurate.

Using Excel to spot check and both the Wolfram and R computer languages to do the bulk of the computations, and based on the Actuarial Value Calculator continuance tables, I have computed the reduction in net claims expenses created by the transitional reinsurance program for 2014, 2015 and 2016. Using the just-increased reinsurance benefits for 2014, the TRP reduced insurers net claims expenses by 14-16% in 2014, depending on the metal level. (Prior to the change last week, the figures were 11-12%, meaning that insurers just received a 3% cash-back rebate from the federal government for 2014.) For 2015, the TRP, assuming its current parameters are not revised, should reduce insurers net claims expenses by 3-4%. And for 2016, the same figures are 3% for all metal levels. Of course, these are average figures. Insurers with unusually large claims expenses may get more benefit out of the TRP. Insurers with unusually low claims expenses may get less.

So, what does his mean. First, since the value of the subsidies has not declined substantially between 2015 and 2016, it is difficult to attribute a substantial part of premium increases to this anticipated change in the subsidy. And even if insurers anticipate some retroactive modification in the generosity of the 2015 program, as has occurred in 2014, I do not see how, in most cases, the 2015-16 phase out of transitional reinsurance would lead to increases on the order of 10% of more. Second, most of the reduction in the TRP occurred between 2014 and 2015. So, the final elimination of the TRP for 2017 should not itself result in enormous increases, though, combined with further increases in claims expenses, might well cast the program deeper into an adverse selection cycle.

A footnote : There has been an implication that the ability of CMS to increase TRP payments for 2014 is a sign that the ACA is working. This is not correct. The main reason TRP payments could increase is that they are proportional to the number of people enrolled.  And because this was at least 14% less that was estimated at the time the original TRP parameters was developed, it is not surprising that, even with higher than expected claims expenses, there could be some extra money to increase the subsidy rate.

# The modification of the Risk Corridors program will usually not be responsible for major increases in gross premiums

Another potential source of premium increases for 2016 is the modification of the Risk Corridors program by the Cromnibus bill.  This is a program that offers a free derivative -- and I mean that in the securities sense --  to insurers participating in the Exchanges.  If they make money -- calculated the Obama administration’s way -- they pay into Risk Corridors, sometimes a substantial amount.  If they lose money, they get paid by Risk Corridors, again, sometimes a substantial amount.  There was no guarantee under the prior program that payments in would equal payments out and work by Standard & Poors indicates that, as some, including me, had earlier predicted, payments out would indeed be greater than payments in.  The Cromnibus bill changes this by making Risk Corridors somewhat akin to bankruptcy.  If obligations out are greater than payments due, the payments out are reduced pro rata until payments out and payments in equilibrate.

So, the question is, to what extent is this change in the law responsible for premium increases that may well be down the pike for 2016.  It is my best estimate that ,as a result of Cromnibus, insurers essentially losing money will receive about 37% of what they would have received had the federal government not required a balanced budget for the program. Of all the things I am testifying about here today, this one, I believe is the one in which there could be significant error bars around my estimate.  There is a lot of information we do not yet know.

I further attempted to estimate how much this increase in downside risk would mean to most insurers. The answer is, it depends.  If, of course, the insurer was pretty confident that it would make money -- something we might ordinarily expect -- then the increase in downside risk does them little harm. Risk corridors only kicks in when you lose money -- or at least are treated as having lost money by the complex formula implemented by CMS. On the other hand, if the insurer thought it would lose money using that formula or it was very uncertain as to what its financial position would be, then the increase in downside risk is somewhat significant.  I therefore estimated that reasonable bounds on the incremental cost to insurers created by Cromnibus ranged from 0 for insurers who expected profitability, 0.5 percent for insurers that thought they would break even and were confident within 95% that their Risk Corridors ratio would range from 0.92 to 1.08, and up to 5% for  an insurer that thought it would lose money but had high uncertainty as to its financial position. I should emphasize that I am looking at the Cromnibus-induced change in Risk Corridors, not at the effect of Risk Corridors as a whole.

The bottom line is, however, when we are looking at gross premium increases over 5% and certainly over 10%, it is unlikely that most of that is the result of the Cromnibus modification of risk corridors.

# The main cause of gross premium increases is likely to be adverse claim experience

This is a conclusion reached partly by a process of elimination. If it' s not the diminution of transitional reinsurance and it is not the Cromnibus modification of Risk Corridors that is responsible for large premium increases, what is it?  While there could conceivably be other factors such as state reguatory developments or interest rate changes, the most obvious candidate is adverse claims experience.  This is particularly so since interest rates have remained relatively stable and the past few years have not been a fertile time for major state regulatory reforms in health insurance. Certainly many of the filings published thus far by insurers seeking gross premium increases in excess of 10% have so stated and work by Standard and Poors strongly indicates that there will far more insurers losing money this year than gaining money.

Before we explore more exotic hypotheses, we should realize that data from the Actuarial Value Calculator that the Obama administration uses to regulated insurers reinforces the belief that adverse claims experience is significantly driving higher premiums.  If one simply looks at the data, the expected claims of an insurer offering a silver policy is 14% higher under that calculator for 2016 than it is for 2015.  The other metals have results of 13%. This data would not factor in either the Transitional Reinsurance Program or the Risk Corridors program.  And lest anyone think there must be something wrong with the data in the Calculator, here is CMS's description of it: "The AV Calculator represents an empirical estimate of the AV calculated in a manner that provides a close approximation to the actual average spending by a wide range of consumers in a standard population. " (https://www.cms.gov/CCIIO/Resources/Regulations-and-Guidance/Downloads/2015-av-calculator-methodology.pdf)

Footnote : I also attempted to see whether there was any difference in the rate of increase between medical claims and pharmaceutical claims. I found no significant difference. This is either because, in fact, there is no difference or because CMS has not yet differentiated in its 2016 drafts of its actuarial value calculator.

---

## Technical Appendix I

# Gross premiums v. Net premiums

In this material, I use the following notation.

- sg1 is the gross premium for the second lowest silver policy in year 1.
- sg2 is the gross premium for the second lowest silver policy in year 2.
- y1 is the subsidy in year 1.
- y2 is the subsidy in year 2.
- rs is the percent increase in the premium for the second lowest silver policy between year 1 and year 2.
- g1 is the gross premium for the policy actually purchased by the insured in year 1.
- rg is the percent increase in the premium for the policy actually purchased by the insured between year 1 and year 2.

If we assume the income of the purchaser remains relatively constant and that the income of the purchaser bounds the most that it can pay for the second lowest silver policy then the subsidy in the second year for the second lowest premium has to be enough so that the net premium remains the same. The algebra below computes a formula for the second year subsidy using this invariance.

```
netPremiumEquivalenceRule = First[Solve[sg1 - y1 == sg2 - y2, y2]] /. sg2 → sg1 (1 + rs)
{y2 → - sg1 + (1 + rs) sg1 + y1}
```

We can now calculate the rate of increase in the net premium if the insured decides to keep its policy (and probably its doctors).

$$\text{netPremiumIncrease} = \frac{\text{g1 (1 + rg) - y2}}{\text{g1 - y1}} - 1;$$

We can simplify this expression using the net premium equivalence rule derived above.

```
simplifiedNetPremiumIncrease =
 Simplify[netPremiumIncrease /. netPremiumEquivalenceRule]
```

$$\frac{\text{g1 rg - rs sg1}}{\text{g1 - y1}}$$

If we stare at this expression for a bit, we can see that it is the difference in the increases in gross premiums divided by the net premium for the policy actually purchased in the prior year.

We can turn this into a function as follows :

$$\text{snpi[g1\_, rg\_, sg1\_, rs\_, y1\_] :=} \frac{\text{g1 rg - rs sg1}}{\text{g1 - y1}}$$

We can test this on various scenarios.  Here is one in which the gross premium in year 1 for the policy chosen is 1000, the rate of gross premium increase in that policy is 10%, the second lowest silver policy has a gross premium of 800 with a rate of gross premium increase of 5%.  And the initial subsidy is 600.

```
snpi[1000, 0.1, 800, 0.05, 600]
0.15
```

We can also evaluate the rate of net premium increase for the policy chosen as a function of the percentage of the premium subsidized, which is a proxy for income.

```
Plot[{snpi[1000, 0.1, 800, 0.05, 800 * subsidyFraction], 0.1},
 {subsidyFraction, 0, 1}, Axes → False, Frame → True, FrameLabel →
  {"subsidy fraction", "net premium increase"}, PlotTheme → "Monochrome",
 PlotLegends → {"net premium increase", "gross premium increase"}]
```




In the interactive version of this document, you can also see a plot such as the one above in which one is permitted to vary other variables such as the ratio of the gross premiums and the two rates of increase.

```
Manipulate[Plot[{snpi[ϵ * 800, rg, 800, rs, 800 * subsidyFraction], rg},
  {subsidyFraction, 0, 1}, Axes → False,
  Frame → True, FrameLabel → {Grid[{{"subsidy fraction"},
      {Row[Riffle[{"⟸", "wealthier", "poorer", "⟹"}, "            "]]}}],
    "net premium increase"}, PlotTheme → "Monochrome",
  PlotLegends → {"net premium increase", "gross premium increase"}],
 {{ϵ, 1.25, "gross premium ratios"}, 1, 2, 0.01, Appearance → "Labeled"},
 {{rg, 0.1, "rate of increase in gross premiums for chosen policy"},
  -0.1, 0.3, 0.01, Appearance → "Labeled"},
 {{rs, 0.05, "rate of increase in gross premiums for second lowest silver"},
  -0.1, 0.3, Appearance → "Labeled"}
]
```




We can also now compute explicitly the circumstances under which the rate of increase in net premiums will exceed the rate of increase in gross premiums. Unfortunately, it is difficult to find a simple English language version of this complex formula.

```
Refine[Reduce[{y1 ≥ 0, g1 > 0, (g1 rg - rs sg1)/(g1 - y1)^2 > rg, g1 ≥ y1, sg1 > 0}, rg, Reals], y1 ≥ 0]
```

$$\left(y1 < g1 < \frac{1}{2}(1 + 2\,y1) + \frac{1}{2}\sqrt{1 + 4\,y1}\ \&\&\ sg1 > 0\ \&\&\ rg > -\frac{rs\,sg1}{-g1 + g1^2 - 2\,g1\,y1 + y1^2}\right) \,||$$
$$\left(g1 == \frac{1}{2}(1 + 2\,y1) + \frac{1}{2}\sqrt{1 + 4\,y1}\ \&\&\ sg1 > 0\ \&\&\ rs < 0\right) \,||$$
$$\left(g1 > \frac{1}{2}(1 + 2\,y1) + \frac{1}{2}\sqrt{1 + 4\,y1}\ \&\&\ sg1 > 0\ \&\&\ rg < -\frac{rs\,sg1}{-g1 + g1^2 - 2\,g1\,y1 + y1^2}\right)$$

## Technical Appendix 2

# The effect of declining transitional reinsurance on insurer prices and exposure under the Affordable Care Act

## Read in data

We begin by reading in data from the 2014, 2015 and 2016 actuarial value calculators.

```
sheets14 = Import[
  "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2014.xlsm",
  "Sheets"]
```

```
{Sheet2, User Guide, Enrollment Restrictions, AV Calculator,
 Variation Results, Platinum Cont. Table - Medical, Gold Cont. Table - Medical,
 Silver Cont. Table - Medical, Bronze Cont. Table - Medical,
 Platinum Cont. Table - Rx Only, Gold Cont. Table - Rx Only,
 Silver Cont. Table - Rx Only, Bronze Cont. Table - Rx Only,
 Platinum Cont. Table - Combined, Gold Cont. Table - Combined,
 Silver Cont. Table - Combined, Bronze Cont. Table - Combined}
```

```
sheets15 = Import[
  "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2015.xlsm",
  "Sheets"]
```

```
{Sheet2, User Guide, Enrollment Restrictions, AV Calculator,
 Variation Results, Platinum Cont. Table - Medical, Gold Cont. Table - Medical,
 Silver Cont. Table - Medical, Bronze Cont. Table - Medical,
 Platinum Cont. Table - Rx Only, Gold Cont. Table - Rx Only,
 Silver Cont. Table - Rx Only, Bronze Cont. Table - Rx Only,
 Platinum Cont. Table - Combined, Gold Cont. Table - Combined,
 Silver Cont. Table - Combined, Bronze Cont. Table - Combined}
```

```
sheets16 = Import[
  "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2016.xlsm",
  "Sheets"]
```

```
{Sheet2, User Guide, AV Calculator,
 Platinum Cont. Table - Medical, Gold Cont. Table - Medical,
 Silver Cont. Table - Medical, Bronze Cont. Table - Medical,
 Platinum Cont. Table - Rx Only, Gold Cont. Table - Rx Only,
 Silver Cont. Table - Rx Only, Bronze Cont. Table - Rx Only,
 Platinum Cont. Table - Combined, Gold Cont. Table - Combined,
 Silver Cont. Table - Combined, Bronze Cont. Table - Combined}
```

## Determine relevant sheets

We  now determine which sheets of these multi - sheet spreadsheets contain the data we want and are common to all sheets.

```
combinedSheets = Select[sheets14 ⋂ sheets15 ⋂ sheets16,
  StringMatchQ[#1, RegularExpression[".+Cont\\.\\s+Table\\s*-\\s*Combined"]] &]
```

```
{Bronze Cont. Table - Combined, Gold Cont. Table - Combined,
 Platinum Cont. Table - Combined, Silver Cont. Table - Combined}
```

```
rxSheets = Select[sheets14 ⋂ sheets15 ⋂ sheets16,
  StringMatchQ[#1, RegularExpression[".+Rx.*"]] &]
```

```
{Bronze Cont. Table - Rx Only, Gold Cont. Table - Rx Only,
 Platinum Cont. Table - Rx Only, Silver Cont. Table - Rx Only}
```

```
medicalSheets = Select[sheets14 ⋂ sheets15 ⋂ sheets16,
  StringMatchQ[#1, RegularExpression[".+Medical.*"]] &]
```

```
{Bronze Cont. Table - Medical, Gold Cont. Table - Medical,
 Platinum Cont. Table - Medical, Silver Cont. Table - Medical}
```

# Main functionality

This section develops the major functions used in the analysis.

```
combinedRegex = RegularExpression[".+Cont\\.\\s+Table\\s*-\\s*Combined"];
```

```
sheetProcess[xlsFileString_, regex_, yearString_] :=
 Module[{s, sheets, sheetsAssoc, dsAssoc, totsAssoc, augAssoc},
  s = Import[xlsFileString, "Sheets"];
  sheets = Select[s, StringMatchQ[#, regex] &];
  sheetsAssoc =
   Association@Map[name ↦ {StringReplace[name, {RegularExpression["\\s+"] → "",
          RegularExpression["\\."] → "", RegularExpression["-"] → ""}],
        yearString} -> Import[xlsFileString,
          {"Sheets", name}][[5 ;; 88, 1 ;; 4]], sheets];
  dsAssoc = Map[sheet ↦ Dataset@Map[row ↦ AssociationThread[
          {"bin", "count", "maxd", "binAverage"} -> row], sheet], sheetsAssoc];
  augAssoc = Map[ds ↦ With[{tot = ds[All /* Total, "count"]},
      ds[All, Append[#, "normalizedCount" → #count / tot] &]], dsAssoc]]
```

```
switchValue[value_, params : {attach_, max_, pct_}] := Position[
   (#1⟦1⟧ ≤ value < #1⟦2⟧ &) /@ Partition[{0, attach, max, ∞}, 2, 1], True, 1, 1]⟦1, 1⟧
```

```
f[aug_, params : {attach_, max_, pct_}] :=
 aug[All /* Total, #normalizedCount Switch[switchValue[#binAverage, params],
     1, #binAverage,
     2, attach + (1 - pct) (#binAverage - attach),
     3, attach + (1 - pct) (max - attach) + (#binAverage - max)] &]
```

```
delta[{x_, y_}, OptionsPattern[{"roundingValue" → 0.01}]] :=
 Round[(x - y)/y, OptionValue["roundingValue"]]
```

# Constants

We input the TRP parameters for each of the years.  We use two parameters for 2014, the original ones and the revised ones.

```
TRPparameters["2014"] = {45 000, 250 000, 0.8};
```

```
TRPparameters["2014a"] = {45 000, 250 000, 1};
```

```
TRPparameters["2015"] = {70 000, 250 000, 0.5};
```

```
TRPparameters["2016"] = {90 000, 250 000, 0.5};
```

```
ignoreReinsurance = {0, 100 000 000, 0};
```

## Plans for 2014

We now process the sheets.

```
sh2014 = sheetProcess[
   "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2014.xlsm",
   combinedRegex, "2014"];
```

And get expected claims for each mental level, with no reinsurance, with the revised reinsurance and with the original reinsurance.

```
noReinsurance2014 = Function[d, f[d, ignoreReinsurance]] /@ sh2014
```

```
⟨|{PlatinumContTableCombined, 2014} → 6153.63,
 {GoldContTableCombined, 2014} → 4995.66, {SilverContTableCombined, 2014} → 4736.52,
 {BronzeContTableCombined, 2014} → 4064.37|⟩
```

```
withRevisedReinsurance2014 = Function[d, f[d, TRPparameters["2014a"]]] /@ sh2014
```

```
⟨|{PlatinumContTableCombined, 2014} → 5406.67,
 {GoldContTableCombined, 2014} → 4337.27, {SilverContTableCombined, 2014} → 4095.71,
 {BronzeContTableCombined, 2014} → 3522.06|⟩
```

```
Merge[{noReinsurance2014, withRevisedReinsurance2014},
 delta[#1, "roundingValue" → 0.01] &]
```

```
⟨|{PlatinumContTableCombined, 2014} → 0.14, {GoldContTableCombined, 2014} → 0.15,
 {SilverContTableCombined, 2014} → 0.16, {BronzeContTableCombined, 2014} → 0.15|⟩
```

## Original Plans for 2014

```
withReinsurance2014 = Function[d, f[d, TRPparameters["2014"]]] /@ sh2014
```

```
⟨|{PlatinumContTableCombined, 2014} → 5556.06,
 {GoldContTableCombined, 2014} → 4468.95, {SilverContTableCombined, 2014} → 4223.87,
 {BronzeContTableCombined, 2014} → 3630.52|⟩
```

```
Merge[{noReinsurance2014, withReinsurance2014}, delta[#1, "roundingValue" → 0.01] &]
```

```
⟨|{PlatinumContTableCombined, 2014} → 0.11, {GoldContTableCombined, 2014} → 0.12,
 {SilverContTableCombined, 2014} → 0.12, {BronzeContTableCombined, 2014} → 0.12|⟩
```

## Plans for 2015

We do the same thing for 2015 ...

```
sh2015 = sheetProcess[
   "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2015.xlsm",
   combinedRegex, "2015"];
```

```
noReinsurance2015 = Function[d, f[d, ignoreReinsurance]] /@ sh2015
```

```
⟨|{PlatinumContTableCombined, 2015} → 6153.63,
 {GoldContTableCombined, 2015} → 4995.66, {SilverContTableCombined, 2015} → 4736.52,
 {BronzeContTableCombined, 2015} → 4064.37|⟩
```

```
withReinsurance2015 = Function[d, f[d, TRPparameters["2015"]]] /@ sh2015
```

```
⟨|{PlatinumContTableCombined, 2015} → 5954.24,
 {GoldContTableCombined, 2015} → 4812.54, {SilverContTableCombined, 2015} → 4556.69,
 {BronzeContTableCombined, 2015} → 3911.76|⟩
```

```
Merge[{noReinsurance2015, withReinsurance2015}, delta[#1, "roundingValue" → 0.01] &]
```

```
⟨|{PlatinumContTableCombined, 2015} → 0.03, {GoldContTableCombined, 2015} → 0.04,
 {SilverContTableCombined, 2015} → 0.04, {BronzeContTableCombined, 2015} → 0.04|⟩
```

## Plans for 2016

And 2016.

```
sh2016 = sheetProcess[
   "/Users/sethchandler/Dropbox/Scholarship/Amsterdam15/avcalculator2016.xlsm",
   combinedRegex, "2016"];
```

### What is the expected payment for 2016 if there were no reinsurance

```
noReinsurance2016 = Function[d, f[d, ignoreReinsurance]] /@ sh2016
```

```
⟨|{PlatinumContTableCombined, 2016} → 6979.6,
 {GoldContTableCombined, 2016} → 5666.2, {SilverContTableCombined, 2016} → 5384.88,
 {BronzeContTableCombined, 2016} → 4602.96|⟩
```

```
withReinsurance2016 = Function[d, f[d, TRPparameters["2016"]]] /@ sh2016
```

```
⟨|{PlatinumContTableCombined, 2016} → 6780.57,
 {GoldContTableCombined, 2016} → 5483.53, {SilverContTableCombined, 2016} → 5204.57,
 {BronzeContTableCombined, 2016} → 4451.09|⟩
```

```
Merge[{noReinsurance2016, withReinsurance2016}, delta[#1, "roundingValue" → 0.01] &]
```

```
⟨|{PlatinumContTableCombined, 2016} → 0.03, {GoldContTableCombined, 2016} → 0.03,
 {SilverContTableCombined, 2016} → 0.03, {BronzeContTableCombined, 2016} → 0.03|⟩
```

## What is the increase in gross expected payments from 2016 relative to 2015

```
Merge[{KeyMap[First, noReinsurance2016], KeyMap[First, noReinsurance2015]},
 delta[#1, "roundingValue" → 0.01] &]
```

```
⟨|PlatinumContTableCombined → 0.13, GoldContTableCombined → 0.13,
 SilverContTableCombined → 0.14, BronzeContTableCombined → 0.13|⟩
```

```
Merge[{KeyMap[First, noReinsurance2016],
  KeyMap[First, noReinsurance2015]}, (#1[[1]] - #1[[2]]) / #1[[2]] &]
```

```
⟨|PlatinumContTableCombined → 0.134225, GoldContTableCombined → 0.134225,
 SilverContTableCombined → 0.136886, BronzeContTableCombined → 0.132514|⟩
```

### What is the expected payment today given 2016 reinsurance

```
withReinsurance2016 = Function[d, f[d, TRPparameters["2016"]]] /@ sh2016
```

```
⟨|{PlatinumContTableCombined, 2016} → 6780.57,
 {GoldContTableCombined, 2016} → 5483.53, {SilverContTableCombined, 2016} → 5204.57,
 {BronzeContTableCombined, 2016} → 4451.09|⟩
```

#### Visualization

```
noReinsurance2016[All, {"binAverage", "normalizedCount"}]
```

```
Missing[KeyAbsent, All]
```

## Technical Appendix 3 : The Risk Corridor Computation

# Post - Cromnibus Risk Corridors

## The original Risk Corridors Formula

We start with the original risk corridors formula and develop a piecewise function that takes x, the risk corridor ratio, and calculates the fraction of that ratio that the government pays to the insurer. If the calculation produces a negative number, the value represents the fraction of that ratio that the insurer pays the government.

$$\text{riskCorridorPayment}[x\_] := \text{Piecewise}\Big[\Big\{\Big\{\frac{1}{2}\left(x-\frac{103}{100}\right),\ \frac{103}{100}<x\leq\frac{108}{100}\Big\},\ \Big\{\frac{9}{10}\left(x-\frac{108}{100}\right)+\frac{1}{2}\left(\frac{108}{100}-\frac{103}{100}\right),\ x>\frac{108}{100}\Big\},\ \Big\{-\frac{1}{2}\left(\frac{97}{100}-x\right),\ \frac{92}{100}\leq x<\frac{97}{100}\Big\},\ \Big\{-\frac{1}{10}\left(9\left(\frac{92}{100}-x\right)\right)-\frac{1}{2}\left(\frac{97}{100}-\frac{92}{100}\right),\ x<\frac{92}{100}\Big\},\ \Big\{0,\ \frac{97}{100}<x<\frac{103}{100}\Big\}\Big\}\Big]$$

```
SetAttributes[riskCorridorPayment, Listable]
```

The plot below maps the risk corridors ratio into the amount of money the government pays under the Risk Corridors program as a function of a particular insurer's risk corridors ratio.

```
Plot[riskCorridorPayment[x], {x, 0.85, 1.15}, PlotTheme → "Detailed", Axes → False,
 Frame → True, FrameLabel → {"risk corridors ratio", "government payment"}]
```




If we assume the risk corridor ratio is normally distributed, which would appear to be a reasonable approximation, we can derive the distribution of risk corridor payments the government is likely to make as a function of m and s, which are respectively the mean of the risk corridor ratio and the standard deviation of the risk corridor ratio. We can write this as $\tau$.

```
τ[m_, s_] :=
 TransformedDistribution[riskCorridorPayment[x], x ≈ NormalDistribution[m, s]]
```

# Computing the Cromnibus Fraction

```
Needs["Notation`"]
```

```
Quiet@Symbolize[ τ⁺ ]; Quiet@Symbolize[ τ⁻ ]
```

Symbolize::bsymbexs : Warning: The box structure attempting to be symbolized has a similar or identical symbol already defined, possibly overriding previously symbolized box structure. >>

Symbolize::bsymbexs : Warning: The box structure attempting to be symbolized has a similar or identical symbol already defined, possibly overriding previously symbolized box structure. >>

The distribution of positive payments can be written as a censored distribution, $\tau^+$, of $\tau$ on the interval $[0,\infty)$. The distribution of negative payments can be written as a censored distribution $\tau^-$ of $\tau$ on the interval $(-\infty,0)$.

```
τ⁺[m_, s_] := CensoredDistribution[{0, ∞}, τ[m, s]]
```

```
τ⁻[m_, s_] := CensoredDistribution[{-∞, 0}, τ[m, s]]
```

We can now calculate the ratio ("the Cromnibus fraction") of $\tau^-$, the amount the government receives, to $\tau^+$, the amount the government is obliged to pay out under Risk Corridors. If the Cromnibus fraction is greater than 1, then, under Cromnibus, all insurers get paid fully. If the Cromnibus fraction is less than 1, however, then, under Cromnibus, the payment the insurer receives is equal to the payment the insurer would have received prior to the Cromnibus bill multiplied by the Cromnibus fraction. It's conceptually not different from figuring out how much unsecured creditors get paid in a bankruptcy: you take the assets of the bankrupt and divide by the liabilities to get the fraction of their claim that each unsecured creditor receives.

We now calculate the mean of the Cromnibus fraction as a function of m and s.

$$\rho[m\_, s\_] := \text{Min}\left[1, \text{Abs}\left[\frac{\text{Mean}[\tau^-[m, s]]}{\text{Mean}[\tau^+[m, s]]}\right]\right]$$

We can derive a formula for $\rho$[m,s].  It is a rather ugly expression and so I will print it out small.

$$\text{Style}[\phi = \rho[m, s], 6]$$

$$
\begin{aligned}
&\text{Min}\Big[1, e^{\text{Re}\left[-\frac{(97-100m)^2}{20000 s^2}+\frac{(103-100m)^2}{20000 s^2}-\frac{(23-25m)^2}{1250 s^2}+\frac{(27-25m)^2}{1250 s^2}\right]} \\
&\quad \text{Abs}\Big[\Big((103-100m)(27-25m)\Big(-853\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m] + 900\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m] - \\
&\qquad 400\sqrt{2}\, e^{\frac{(97-100m)^2}{20000 s^2}}\, s\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m] - 500\sqrt{2}\, e^{\frac{(23-25m)^2}{1250 s^2}}\, s\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m] - \\
&\qquad 485\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{97-100m}{100\sqrt{2}\, s}\right] - 368\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{23-25m}{25\sqrt{2}\, s}\right] - \\
&\qquad 900\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{m}{\sqrt{2}\, s}\right] + 48500\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[23-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[97-100m]}{100\sqrt{2}\, s}\right] - \\
&\qquad 50000\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[23-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[97-100m]}{100\sqrt{2}\, s}\right] + 9200\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[23-25m]}{25\sqrt{2}\, s}\right] - \\
&\qquad 10000\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[23-25m]}{25\sqrt{2}\, s}\right] + 900\, e^{\frac{(97-100m)^2}{20000 s^2}+\frac{(23-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[97-100m]\, \text{Abs}[23-25m]\, \text{Erf}\left[\frac{\text{Abs}[m]}{\sqrt{2}\, s}\right]\Big)\Big) \Big/ \\
&\quad \Big(\text{Abs}[97-100m]\, \text{Abs}[23-25m]\Big(900\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m] + 400\sqrt{2}\, e^{\frac{(103-100m)^2}{20000 s^2}}\, s\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m] + \\
&\qquad 500\sqrt{2}\, e^{\frac{(27-25m)^2}{1250 s^2}}\, s\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m] + 515\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{103-100m}{100\sqrt{2}\, s}\right] - \\
&\qquad 515\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{27-25m}{25\sqrt{2}\, s}\right] + 900\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{m}{\sqrt{2}\, s}\right] - \\
&\qquad 51500\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[103-100m]}{100\sqrt{2}\, s}\right] + 50000\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[103-100m]}{100\sqrt{2}\, s}\right] - \\
&\qquad 10800\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[27-25m]}{25\sqrt{2}\, s}\right] + 10000\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[m]\, \text{Erf}\left[\frac{\text{Abs}[27-25m]}{25\sqrt{2}\, s}\right] - \\
&\qquad 900\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}}\, m^2 \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Erf}\left[\frac{\text{Abs}[m]}{\sqrt{2}\, s}\right] - 947\, e^{\frac{(103-100m)^2}{20000 s^2}+\frac{(27-25m)^2}{1250 s^2}} \sqrt{\pi}\, \text{Abs}[103-100m]\, \text{Abs}[27-25m]\, \text{Abs}[m]\, \text{Erfc}\left[\frac{27-25m}{25\sqrt{2}\, s}\right]\Big)\Big)\Big]\Big]
\end{aligned}
$$

# Estimating the parameters to the Cromnibus Fraction

We now have a general  formula that determines, for any mean value and standard deviation value of the risk corridors ratio, the Cromnibus Fraction, the percentage of payments that the government will now make to insurers losing money prior to the Cromnibus bill, to the payments that would have been owing prior the passage of Cromnibus.

To undertake this computation, we use information from Standard and Poors, which recently used data to estimate the percentage of insurers (14%) that would be receiving money under Risk Corridors and the percentage of insurers (30%) that would be losing money. Assuming again that the risk corridors ratio is normally distributed, this devolves into an algebra problem of two formulas and two unknowns. The list of rules $\alpha$ holds the solution.

```
α =
 N[ToRules[Refine[Reduce[{m ∈ Reals, s ∈ Reals, CDF[NormalDistribution[m, s], 97/100] ==
      14/100, SurvivalFunction[NormalDistribution[m, s], 103/100] == 30/100},
     {m, s}, Reals], (Erfc[(103/100 - (97 InverseErfc[3/5] + 103 InverseErfc[7/25])/(100 InverseErfc[3/5] + 100 InverseErfc[7/25]))/(Sqrt[2] s)] |
      Erfc[((97 InverseErfc[3/5] + 103 InverseErfc[7/25])/(100 InverseErfc[3/5] + 100 InverseErfc[7/25]) - 97/100)/(Sqrt[2] s)]) ∈ Reals]]]
```

```
{m → 1.01039, s → 0.0373897}
```

# Calculating the most likely "Cromnibus Fraction"

So, it turns out that the mean value of the distribution of risk corridors ratio that can derived from the Standard and Poors data is 1.01039 and the standard deviation of the distribution of the risk corridors ratio that can be similarly be derived is 0.0373897. The probability density function of the risk corridors ratio implied by the Standard and Poors data thus looks as follows.

```
Plot[PDF[NormalDistribution[m, s] /. α, x], {x, 0.8, 1.2},
 PlotTheme → "Detailed", FrameLabel → {"risk corridor ratio", "PDF"}]
```




This calculation now allows us readily to determine the most probable Cromnibus Fraction.

```
ϕ /. α
```

```
0.366729
```

So, as a result of Cromnibus, insurers should expect to receive roughly 37 % of what they would have received prior to Cromnbus. And they should expect to pay the same amount as they did if they make money.

Here is now a revised graph of what the government will pay as a function of the risk corridors ratio.

```
postCromnibusRiskCorridorPayment[x_] :=
 If[x < 1, riskCorridorPayment[x], (ϕ /. α) riskCorridorPayment[x]]
```

```
Plot[postCromnibusRiskCorridorPayment[x],
 {x, 0.85, 1.15}, PlotTheme → "Detailed", Axes → False, Frame → True,
 FrameLabel → {"risk corridors ratio", "government payment"}]
```




We can also compare the position of the insurer post-Cromnibus to what it would have been if there were no Risk Corridors at all and what it would have been if the government had paid what the statute said was owed prior to the passage of Cromnibus.

```
Plot[{1 - x, 1 - (x - riskCorridorPayment[x]),
  1 - (x - postCromnibusRiskCorridorPayment[x])},
 {x, 0.85, 1.15}, PlotTheme → "Detailed", Axes → False, Frame → True,
 FrameLabel → {"risk corridors ratio", "insurer position"},
 PlotLegends → {"no Risk Corridors",
   "pre-Cromnibus Risk Corridors", "post-Cromnibus Risk Corridors"}]
```




# Valuing the greater risk to insurers post - Cromnibus

Suppose a given insurer believes its risk corridors ratio is distributed as a normal distribution with mean

1 and standard deviation 0.04.

```
𝒩A = NormalDistribution[1., 0.04]
```

```
NormalDistribution[1., 0.04]
```

Without risk corridors the distribution of (roughly) profits is then

```
d0 = TransformedDistribution[1 - x, x ≈ 𝒩A]
```

```
NormalDistribution[0., 0.04]
```

We can now calculate the expected position of the insurer if they are risk averse.

```
ρ0 = Quiet[With[{e0 = SmoothKernelDistribution[Sort[RandomVariate[d0, 10 000]]]},
   NExpectation[Quantile[e0, q], q ≈ BetaDistribution[1, 2]]]]
```

```
-0.0224289
```

With original risk corridors, the distribution (again, roughly) of profits is ...

```
d1 = TransformedDistribution[1 - (x - riskCorridorPayment[x]), x ≈ 𝒩B]
```

TransformedDistribution[

$$1 - \dot{x} + \left( \begin{cases} \frac{1}{2}\left(-\frac{103}{100} + \dot{x}\right) & \frac{103}{100} < \dot{x} \le \frac{27}{25} \\ \frac{1}{40} + \frac{9}{10}\left(-\frac{27}{25} + \dot{x}\right) & \dot{x} > \frac{27}{25} \\ \frac{1}{2}\left(-\frac{97}{100} + \dot{x}\right) & \frac{23}{25} \le \dot{x} < \frac{97}{100} \\ -\frac{1}{40} - \frac{9}{10}\left(\frac{23}{25} - \dot{x}\right) & \dot{x} < \frac{23}{25} \\ 0 & \text{True} \end{cases} \right), \ \dot{x} \approx \text{NormalDistribution}[1.1, 0.08]]$$

We can now calculate the expected position of the insurer if they are risk averse.

```
ρ1 = Quiet[With[{e1 = SmoothKernelDistribution[Sort[RandomVariate[d1, 10 000]]]},
   NExpectation[Quantile[e1, q], q ≈ BetaDistribution[1, 2]]]]
```

```
-0.0592451
```

With post-Cromnibus risk corridors, the distribution (again, roughly) of profits is ...

```
d2 = TransformedDistribution[1 - (x - postCromnibusRiskCorridorPayment[x]), x ≈ 𝒩A]
```

```
TransformedDistribution[
 1 - ẋ + If[ẋ < 1, riskCorridorPayment[ẋ], (ϕ /. α) riskCorridorPayment[ẋ]],
 ẋ ≈ NormalDistribution[1., 0.04]]
```

Finally, we can calculate the expected position of the insurer if they are risk averse.

```
ρ2 = With[{e2 = SmoothKernelDistribution[Sort[RandomVariate[d2, 10 000]]]},
  NExpectation[Quantile[e2, q], q ≈ BetaDistribution[1, 2]]]
```

```
-0.0216643
```

```
ρ0 - ρ2
```

```
-0.000764594
```

What we can see is that if the insurer is moderately risk averse but expected to break even on plans sold on an Exchange, the original risk corridors saved them about 0.56% of their expenses. Note that this is 0.56%, not 5.06%. The post-Cromnibus risk corridors will save them only 0.16% of their expenses. Thus the Cromnibus modification of risk corridors does not have in the average case what most would consider a large effect on probable insurer pricing.

There is an exception worth discussing. Consider the insurer who, because they wanted to bring people into their network was willing to price polices such that the distribution of the risk corridors ratio was as follows:

```
𝒩B = NormalDistribution[1.1, 0.08]
NormalDistribution[1.1, 0.08]
```

This reflects higher expenses but also more uncertain expenses.

We can now rerun the computations under this assumption.

Without risk corridors the distribution of (roughly) profits is then

```
d0B = TransformedDistribution[1 - x, x ≈ 𝒩B]
NormalDistribution[-0.1, 0.08]
```

We can now calculate the expected position of the insurer if they are risk averse.

```
ρ0B = Quiet[With[{e0B = SmoothKernelDistribution[Sort[RandomVariate[d0B, 10 000]]]},
   NExpectation[Quantile[e0B, q], q ≈ BetaDistribution[1, 2]]]]
-0.145109
```

With original risk corridors, the distribution (again, roughly) of profits is ...

```
d1B = TransformedDistribution[1 - (x - riskCorridorPayment[x]), x ≈ 𝒩B]
TransformedDistribution[
```

$$1 - \dot{x} + \begin{cases} \frac{1}{2}\left(-\frac{103}{100} + \dot{x}\right) & \frac{103}{100} < \dot{x} \le \frac{27}{25} \\ \frac{1}{40} + \frac{9}{10}\left(-\frac{27}{25} + \dot{x}\right) & \dot{x} > \frac{27}{25} \\ \frac{1}{2}\left(-\frac{97}{100} + \dot{x}\right) & \frac{23}{25} \le \dot{x} < \frac{97}{100} \\ -\frac{1}{40} - \frac{9}{10}\left(\frac{23}{25} - \dot{x}\right) & \dot{x} < \frac{23}{25} \\ 0 & \text{True} \end{cases}, \ \dot{x} \approx \text{NormalDistribution}[1.1, 0.08]]$$

We can now calculate the expected position of the insurer if they are risk averse.

```
ρ1B = Quiet[With[{e1B = SmoothKernelDistribution[Sort[RandomVariate[d1B, 10 000]]]},
   NExpectation[Quantile[e1B, q], q ≈ BetaDistribution[1, 2]]]]
-0.0589978
```

With post-Cromnibus risk corridors, the distribution (again, roughly) of profits is ...

```
d2B = TransformedDistribution[1 - (x - postCromnibusRiskCorridorPayment[x]), x ≈ 𝒩B]
TransformedDistribution[
 1 - ẋ + If[ẋ < 1, riskCorridorPayment[ẋ], (ϕ /. α) riskCorridorPayment[ẋ]],
 ẋ ≈ NormalDistribution[1.1, 0.08]]
```

Finally, we can calculate the expected position of the insurer if they are risk averse.

```
ρ2B = With[{e2B = SmoothKernelDistribution[Sort[RandomVariate[d2B, 10 000]]]},
  NExpectation[Quantile[e2B, q], q ≈ BetaDistribution[1, 2]]]
```

-0.114678

```
ρ0B - ρ1B
```

-0.0861109

```
ρ0B - ρ2B
```

-0.030431

In this scenario, the original risk corridors saved the insurer about 8.6 % of their expenses; the post-Cromnibus risk corridors program saved the insurer about 3.1% of their expenses.  This would be a practical upper bound on the effect of risk corridors.